EE230 : Analog Circuits Lab

Mayur Ware | 19D070070, **Section 6**

Experiment 8: Special Opamp Linear Circuits

October 1, 2021

Part A – Precision Rectifiers

Improved Half-Wave Rectifier-A




```
Mayur Ware | 19D070070 | Improved Half Wave Rectifier-A
.include ua741.txt
.include IN914.txt
*Netlist
R1 In Neg 10k
R2 Neg Out 10k
Rl Out GND 1k
D1 Neg O1 1N914
D2 O1 Out 1N914
X1 GND Neg PP NP O1 ua741
VCCp PP GND 12
VCCm NP GND -12
Vin In GND sin(0 1 100 0 0 0)
*Analysis
.tran 1u 40m
*Control Commands
.control
run
plot V(Out) V(In)
plot V(Out) vs V(In)
.endc
.end
```

Plots :



V_{in} and V_{out}



V_{out} vs V_{in}

Learning :

I learned that with using the Improved Half Wave Rectifier A/B we can solve the slow speed of Half Wave Rectifier by avoiding the problematic clipping at negative V_{in}

Full-Wave Rectifier



Full-wave rectifier

```
Mayur Ware | 19D070070 | Full Wave Rectifier
.include ua741.txt
.include IN914.txt
*Netlist
R1 In Neg1 10k
R2 Neg1 In1 10k
Rs1 In Neg2 10k
Rs2 In1 Neg2 5k
Rf Neg2 Out 10k
Rl Out GND 1k
D1 O1 Neg1 1N914
D2 In1 O1 1N914
X1 GND Neg1 PP NP O1 ua741
X2 GND Neg2 PP NP Out ua741
VCCp PP GND 12
VCCn NP GND -12
Vin In GND sin(0 1 100 0 0 0)
*Analysis
.tran 1u 40m
*Control Commands
.control
run
plot V(Out) V(In)
plot V(Out) vs V(In)
.endc
.end
```

Plots :



V_{in} and V_{out}



V_{out} vs V_{in}

Learning :
I learned that how Full Wave Rectifier is different from Half Wave Rectifier in working and functioning

Single-pole Active High-pass Filter




```
Mayur Ware | 19D070070
*Single Pole Active Highpass Filter
.include ua741.txt
*Netlist
R1 Out Neg 9.1k
R2 Neg GND 1k
Ra Pos GND 4.7k
Ca In Pos 0.1u
X1 Pos Neg PP NP Out ua741
VCCp PP GND 12
VCCn NP GND -12
Vin in gnd dc 0 ac 1
*Analysis
.ac dec 100 1 10k
.control
run
set color0 = white
set color1 = black
set color2 = blue
set color3 = red
plot vdb(Out)
.endc
.end
```

Plots :



Frequency Response of Single-pole Active High-pass Filter

Simulation :
The cut-off frequency is given by $fc = 1/2\pi R_A C_A = 338.638$Hz. The slope of the curve is +20 dB before corner frequency and 0dB after it. We can verify it from the graph as well.

Learning :
The frequency response of the circuit is the same as that of the passive filter, except that the amplitude of the signal is increased by the gain of the amplifier and for a non-inverting amplifier the value of the pass band voltage gain is given as 1 + R2/R1

Sallen-Key (2-pole) Active Low-pass Filter




```
Mayur Ware | 19D070070
*Sallen-Key (2-pole) Active Lowpass Filter
.include ua741.txt
*Netlist
R1 Out Neg 1.8k
R2 Neg GND 3.3k
Ra In A 4.7k
Rb A Pos 4.7k
Ca A Out 0.1u
Cb Pos GND 0.1u
X1 Pos Neg PP NP Out ua741
VCCp PP GND 12
VCCn NP GND -12
Vin In GND dc 0 ac 1
*Analysis
.ac dec 100 1 10k
.control
run
plot vdb(Out)
.endc
.end
```

Plots :



Frequency Response of Sallen-Key (2-pole) Active Low-pass Filter

Simulation :
The cut-off frequency is given by $fc = 1/2\pi\sqrt{R_A R_B C_A C_B}$ = 338.638Hz. The slope of the curve is +0 dB before corner frequency and -40 dB after it. We can verify it from the graph as well.

Learning :
I learned the working and functioning of Sallen-Key (2-pole) Active Low-pass Filter using NGSpice and beyond the corner frequency, the filter attenuates the frequency response

References

1) Lecture Slides
2) Sedra-Smith
3) WEL Resources for NGSpice